

Estimation of Adherence to Antipsychotic and Diabetic Medications in a Sample of Schizophrenia Patients

Venky Chakravarthy PhD, BioPharma Data Services, Ann Arbor, MI
Thomas N. Taylor, PhD, Wayne State University, Detroit, Michigan

ABSTRACT

We studied a sample of Schizophrenia patients and were interested in knowing (1) Are schizophrenia patients adherent to their medication if they are also taking diabetic medication? And (2) How adherent to their medication are Schizophrenia patients who have no other co-morbidity and are their significant differences between these two groups? (3) Can we work with a simplified data structure using multiple records per patient and attain efficiency while maintaining accuracy?

The results indicate that Schizophrenia patients with no co-morbidity are less likely to be adherent to their medication than patients on multiple medications. However, this difference was not statistically significant.

We were able to work with a simplified data structure and were able to achieve efficiency in calculations using advanced techniques to calculate the medication possession ration (MPR).

INTRODUCTION

Non adherence to medication is a significant health care cost. Roughly a quarter of the patients on medication are non-compliant (Bovet et al., 2002). The New England Health Care Institute (NEHI) has recently estimated the annual cost of medication non-adherence at \$290 Billion (August 11, 2009). The non-adherence issue is recognized as a significant problem by the International Society for Pharmacoeconomics and Outcomes Research (ISPOR). It has set up special task groups who have initiated studies to seek a greater understanding of the problem.

BRIEF REVIEW OF THE LITERATURE

While there are extensive studies on the medication adherence problem and the economic costs associated with it, the method of estimation has received limited coverage in the SAS literature. Estimating medication adherence is a complicated task. Sophisticated algorithms have been developed to capture the Medication Possession Ratio (MPR) which is a measure of medication adherence (Leslie, 2007). However, medication adherence has been primarily examined from the point of view of a single drug. What happens when the patient is on multiple drugs?

There is recent evidence that patients on single medication for a chronic condition are less adhering to medication than those on multiple medications (Kreyenbuhl et al., 2010).

We will briefly touch upon the study design, methods and results which may be published in detail elsewhere later. The main focus of this paper will be on the SAS® use for our study. We present a different approach than using array manipulation for Medical Possession Ratio calculations. We explain the merits of using a narrow and long dataset as opposed to a wide data set typically used for extensive array manipulation. We also briefly demonstrate the efficiencies gained using the hash data step object for looking-up subsets. Finally, we present results using re-sampling of the data to guard against spurious results (Wesfall, 1999).

STUDY DESIGN

The MarketScan® Commercial Claims & Encounters data for years 2006 and 2007 were restricted to a sample of Schizophrenia patients, <65 years of age, who were also on medication for diabetes control. The data were cleaned for quality control.

We hypothesize based on the recent study that Schizophrenia patients on single medication are less likely to be adherent to their medication than those with co-morbidity and on multiple medications. Therefore, we expect to see lower adherence among Schizophrenia patients who were not on any other chronic medication (n=444) compared to those who were also on Diabetic control medication (n=864). We define not being on any other chronic medication as follows. The number of days in possession of any non-antipsychotic medication must not exceed one quarter of the number of days in possession of an antipsychotic medication in a year.

SELECTION OF PATIENTS

We filtered the patients in the following order.

1. Schizophrenia cases: patients with at least one inpatient or outpatient medical claim with a diagnosis of schizophrenia (.ICD-9 codes 295.xx) and at least one prescription for an antipsychotic medication in 2006. Cases were further restricted to those individuals participating in health plans that reported prescription drug data for both 2006 and 2007.
2. Subset who were on Diabetic medication (had at least one prescription for a diabetes medication in 2006)
3. Subset who were NOT on Diabetic medication (mutually exclusive from 2)
4. Subset Diabetic patients to those who also took Antipsychotic medication.
5. Subset NON-Diabetic patients to those who took Antipsychotic medication and no other chronic medication.

Since these are large data sets we had to be mindful of efficient selection processes. Given below are the salient parts of the code with brief description headers.

```

%*-----;
%*   Assign values to a macro variable for repeated KEEP statements   ;
%*-----;
%let keeplist = enrolid svcdate ndcnum daysupp thercls ;

%*-----;
%*   Extract Schizophrenia patients taking diabetic medication       ;
%*-----;
data diablister ;
  set schizo.ccaed062_schizo (keep=&keeplist)
      schizo.ccaed071_schizo (keep=&keeplist)
      ;
  by enrolid ;
  where not missing(enrolid) and
        sign(daysupp) >-1 and
        thercls in (172,173,174) ;
run;

%*-----;
%*   Get a list of the IDs to load to a memory table to extract     ;
%*   subsets.                                                       ;
%*-----;
proc sql noprint ;
  create table diabidlist as
  select distinct enrolid
  from diablister ;
  select count(distinct enrolid)
  into : idcount separated by " "
  from diablister ;
quit ;
%put <&idcount> ;

%*-----;
%*   Extract patients taking antipsychotic medications             ;
%*   limiting them to the subset taking diabetic medication.       ;
%*   The Data Step Hash Object is a good candidate for efficiency  ;
%*   Define the Hash objects on the first iteration of the datastep. ;
%*-----;
```

```

data psychanddiab ;
  if _N_ = 1 then do;
    declare Hash ht(dataset: "diabidlist"); -----(1)
    ht.defineKey("enrolid"); -----(2)
    ht.defineData("enrolid"); -----(3)
    ht.defineDone(); -----(4)
  end;

  /* Load enrolid from both and query hash table */
  /* This should give all patients who took THERCLS=70 */
  set schizo.ccaed062_schizo (keep=&keeplist)
      schizo.ccaed071_schizo (keep=&keeplist) ;
  by enrolid ;
  where not missing(enrolid) and
        sign(daysupp) >-1 and
        thercls = 70 ;
  if ht.find() = 0; *patients in IDLIST; -----(5)
run;

```

A brief explanation about the Hash object used above. This is a relatively new feature in SAS introduced in Version 9. When merging two datasets, if one dataset is small compared to other, the entire small dataset can be loaded to memory and looked up by the same key of the larger dataset. This makes the process much quicker. There were a number of papers using the hashing method with temporary arrays even before the introduction of the hash object (e.g. Dorfman, 2001). In the code above the reference line (1) declares the DIABIDLIST dataset as the one to load in memory. The key on which the larger dataset(s) will search is declared in (2) as ENROLID. The same variable is declared as satellite data to be carried with it to the output dataset PSYCHANDDIAB. Finally (4) says we have done defining our hash table. Now the job of the larger dataset is to go find a match on ENROLID which is stated in (5). When the ID is found a return code of 0 is returned indicating success and the record is output to the dataset PSYCHANDDIAB. There are many complex pieces to the Hash object and the reader is referred to a nice introduction by Secosky and Bloom (2006).

SELECTION OF COMPARISON PATIENTS

The Schizophrenia patients with diabetes were compared to Schizophrenia patients without diabetes or any other chronic medication use. We had operationally defined this as no other medication should be filled more than 25 percent of the days that antipsychotic medications were filled for that patient. If exceeded that patient is dropped. So this could be a tricky selection but could be simplified keeping the same data structure with multiple rows per patient. However, a temporary array is used to load the days supplied (DAYSUPP) that was summed by therapeutic class (THERCLS) and given a new name SUMDAYSUPP. Our reference SUMDAYSUPP is therapeutic class value 70 for antipsychotic medication. Once the data step reads all the records for the patient, we examine every element of the array at the end of the patient's record. Please note that this type of DO loop with a SET inside it is not conventional use. However, we can take advantage of certain desirable properties to limit the number of steps in the coding. For a more complete discussion of this DO loop structure, see Chakravarthy (2003) and Dorfman and Vyverman (2009).

```

%*-----;
%* Sum up the days supplied by enrolid by thercls. This allows ;
%* us to examine later for other medications for chronic use. ;
%*-----;
data checkrestpsych ;

  array lup (300) _temporary_ ;
  /* load all DAYSUPP by enrolid */
  do until (last.enrolid) ;
    set restpsych ;
    by enrolid thercls ;
    lup(thercls) = sumdaysupp ;
  end;

```

```

    if thercls = 70 then reference = sumdaysupp ;
end ;

/* initialize for selection by default and mark for
deletion only when the deletion criteria is met
array element with index value 70 is antipsychotic
which is the reference value. This should not be
evaluated */
todelete = 0 ;
do _n_ = 1 to dim(lup) ;
  if not todelete then
  do ;
    if not missing(lup(_n_)) and _n_ ^=70 and
      (lup(_n_) > (reference) or
      lup(_n_) > (.25*reference))
    then do ;
      todelete = 1 ;
      tocheckdaysupp = lup(_n_) ;
      tocheckthercls = _n_ ;
    end ;
  end ;
  lup(_n_) = . ;
end ;
run ;

```

CALCULATION OF THE MEDICATION POSSESSION RATIO (MPR)

Once the assembling of the data was completed, it was time to calculate the medication possession ration (MPR). We used the ISPOR definition:

Number of Days of Medication Supplied within the Refill Interval/Number of Days in Refill Interval.

In past SAS papers, the use of ARRAYS was a popular choice for this calculation. In this study we have used logic similar to that but have worked with the simpler data structure of multiple records per patient. This is easier to view when printed or seen through a VIEWTABLE window. ARRAYS are still used but only for populating the calculated MPR values to the different Medications. Here is the single datastep that calculates MPR.

```

data mpr5 ;
  do until (last.name) ;
    set temp5_3 ;
    by enrolid name svcdate;
    /* once the refill date goes past a year from
    first prescription we want to stop
    processing. The CONTINUE statement
    continues with the DO loop without
    doing anything further.*/
    if stopProcessDate then continue ;
    if first.name then
      do ;
        cumdays = daysupp ;
        totdays = daysupp ;
        firstSupplyDate = svcdate ;
        /* This is the cutoff date */
        AnniversaryDate = ( firstSupplyDate + 365 ) - 1 ;
      end ;
    if svcdate >= AnniversaryDate

```

```

then stopProcessDate = svcdate ;
if stopProcessDate then continue ;

if not first.name and
  not (first.name and last.name) and
  (svcdate - PrevServiceDate > (.5*PrevSupply))
then cumdays = sum(cumdays,daysupp) ;
PrevServiceDate = svcdate ;
PrevSupply = daysupp ;
end ;

array allrx (*) &druglist ;
array psych (*) &druglist2 ;
retain allrx psych ;

if stopProcessDate
then cumdays = cumdays - prevsupply +
  (AnniversaryDate-PrevServiceDate+1) ;

if not cumdays then delete ;
allrx(input(name,indrug.)) = cumdays / 365 ;
if name in (&psychdrugs) then
psych(input(trim(name)||"2",indrug.)) = cumdays ;

/* This calculates the MPR for all of ATYPICAL and non-ATYPICAL
antipsychotics combined */
if last.enrolid then
do ;
  psychMPR = (sum(of psych(*))) /
    ((dim(psych)-nmiss(of psych(*))) * 365) ;
  output ;
  do _n_ = 1 to dim (allrx) ;
    allrx(_n_) = . ;
  end ;
  do _n_ = 1 to dim (psych) ;
    psych(_n_) = . ;
  end ;
end ;
format AnniversaryDate PrevServiceDate firstSupplyDate
  stopProcessDate mmddyy10. ;
keep enrolid ndcnum svcdate daysupp prodnme name
  AnniversaryDate PrevServiceDate firstSupplyDate
  stopProcessDate cumdays &druglist &druglist2 psychMPR
  PrevSupply ;
run ;

```

METHODS AND DISCUSSION OF RESULTS

The descriptive statistics of the demographics in the study is produced below in tables 1 and 2.

Table 1: Descriptive Statistics on Age Distribution

Age of Patient				
N	Mean	Std Dev	Minimum	Maximum
1308	45.1437309	12.8874529	10.0000000	63.0000000

Table 2: Frequency Distribution of Gender

Gender of Patient				
SEX	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1- M	606	46.33	606	46.33
2- F	702	53.67	1308	100.00

The MPR ratios were calculated individually for each Atypical and Non-Atypical medication. The oral diabetic medications were combined and the Insulin class was separated. The mean Medication Possession Ratio is displayed in Table 3.

Table 3: Mean Medication Possession Ratios

Analysis Variable : Medication Possession Ratio					
Name of Medication	N Obs	Lower 95% CL for Mean	Upper 95% CL for Mean	Mean	Std Dev
ABILI	266	0.5360	0.6194	0.5777	0.3453
CLOZA	87	0.7036	0.8304	0.7670	0.2974
GEODO	148	0.5228	0.6398	0.5813	0.3601
INVEG	27	0.1657	0.3167	0.2412	0.1909
RISPE	282	0.6021	0.6793	0.6407	0.3293
SEROQ	275	0.5749	0.6564	0.6156	0.3433
ZYPRE	143	0.5594	0.6709	0.6151	0.3373
OTHER	288	0.5683	0.6473	0.6078	0.3407
INSULIN	238	0.5198	0.5980	0.5589	0.3059
ORALDIAB	795	0.6905	0.7322	0.7114	0.2996
All Antipsychotics	864	0.6501	0.6881	0.6691	0.2845
Antipsychotics No Co-morbidity	444	0.5899	0.6491	0.6195	0.3173

We were interested in knowing how adherent are Schizophrenia patients with (a) their antipsychotic medications and (b) their diabetic medication. We were also interested in knowing how adherent to medications are Schizophrenia patients having no other co-morbidity. Based on a recent publication, patients on single medication are less likely to be adherent than those using multiple drugs. Preliminary indication points to the same. Patients on single medication appear to be less likely to be adherent to their medication than patients with co-morbidity and on multiple chronic medications. However, this needs to be tested for statistical significance. We follow Pearce and Wesfall (1998) recommendation and use PROC MULTTEST resampling without replacement.

Table 4: Test of Differences in Mean Medication Possession Ratios

p-Values			
Variable	Contrast	Raw	Stepdown Permutation
Medication Possession Ratio	OralDiab vs Psych	0.0069	0.0137
	OralDiab vs PsychONLY	<.0001	<.0001
	Psych vs PsychONLY	0.1280	0.1278

Even though preliminary indication was that schizophrenic patients with no co-morbidity were less likely to be adherent to their medications than patients on multiple chronic medications, the difference is not statistically significant. It is to be noted that a test of the mean differences using the same MULTTEST procedure yielded a statistically significant differences between the two groups. However, we followed Pearce and Westfall's recommendation to use the Cochran-Armitage linear trend tests for group comparisons. Patients in the study were more likely to be adherent to their diabetic medication than their antipsychotic medication. This was statistically significant.

CONCLUSION

We had hypothesized that Schizophrenia patients with no other co-morbidity are less likely to adhere to their medication than patients who were on multiple medications. This logic was based on the results from a recent publication. Although this appeared true in our study the difference was not statistically significant.

The main focus of this paper was to demonstrate efficiencies attained in using certain SAS code. We also demonstrated a different method of calculating MPR using the standard definition. In the process we showed how a narrow and long data structure is more easy to handle than a wide and short data structure.

REFERENCES

- J. Kreyenbuhl, L. B. Dixon, J. F. McCarthy, S. Soliman, R. V. Ignacio, and M. Valenstein (2010). Does Adherence to Medications for Type 2 Diabetes Differ Between Individuals With Vs Without Schizophrenia? *Schizophr Bull*, March 1, 2010; 36(2): 428 - 435.
- Pascal Bovet, Michel Burnier, George Madeleine, Bernard Waeber, and Fred Paccaud (2002). **Monitoring one-year compliance to antihypertension medication in the Seychelles.** http://whqlibdoc.who.int/bulletin/2002/Vol80-No1/bulletin_2002_80%281%29_33-39.pdf
- NEHI (2009). **Thinking Outside the Pillbox: A System-wide Approach to Improving Patient Medication Adherence for Chronic Disease.** http://www.nehi.net/news/press_releases/110/nehi_research_shows_patient_medication_nonadherence_costs_health_care_system_290_billion_annually
- Scott, Leslie (2007). **Using Arrays to Calculate Medication Utilization.** SAS Global Forum. <http://www2.sas.com/proceedings/forum2007/043-2007.pdf>
- Westfall, Peter H. (1999) **Advances in Multiple Comparisons and Multiple Tests using the SAS® System.** SAS Institute Inc., *Proceedings of the Twenty-Fourth Annual SAS® Users Group International Conference.* Cary, NC: SAS Institute Inc
- SAS Institute Inc. 2010. *SAS 9.1.3 Base SAS Language Reference.* <http://support.sas.com/onlinedoc/913/docMainpage.jsp> (accessed March 04, 2010).
- Dorfman, Paul M. (2001). **Table Look-Up by Direct Addressing: Key-Indexing -- Bitmapping – Hashing.** SAS Institute Inc., *Proceedings of the Twenty-Sixth Annual SAS® Users Group International Conference.* Cary, NC: SAS Institute Inc

Dorfman, Paul M. and Vyverman, Koen (2009). **The DOW-Loop Unrolled**. SAS Institute Inc. *Proceedings of the SAS® Global Forum 2009 Conference*. Cary, NC: SAS Institute Inc

Chakravarthy, Venky (2003). **The DOW (not that DOW!!!) and the LOCF in Clinical Trials**. SAS Institute Inc., *Proceedings of the Twenty-Eighth Annual SAS® Users Group International Conference*. Cary, NC: SAS Institute Inc

Secosky and Bloom (2006). **Getting Started with the DATA Step Hash Object**. SAS Institute Inc., *Proceedings of the Two thousand Six Annual South Central SAS® Users Group Conference*. Cary, NC: SAS Institute Inc

Pearce, Gregory L. and Westfall, Peter H. (1998). **Using Resampling Techniques in PROC MULTTEST to Evaluate Surgeon Specific Results Following Coronary Artery Bypass Graft (CABG) Surgery**

ACKNOWLEDGMENTS

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Venky Chakravarthy
Enterprise: BioPharma Data Services
E-mail: Venky@biopharmadataservices.com
Web: <http://www.biopharmadataservices.com>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.